11 Lock-In Verstärker

Markus Lippitz

16. Juni 2022

Ziele Sie können *erklären*, wie ein Lock-in Verstärker funktioniert und ihn zur Messung kleiner Signal *benutzen*.

- Lock-in Verstärker
- Seitenbänder bei Amplituden-Modulation
- Boxcar averager

Literatur Horowitz/Hill Kap. 15.12–15, Tutorial von Bentham Instruments

Überblick

Ein Lock-In-Verstärker (engl. lock-in amplifier, LIA) sollte besser phasen-empfindlicher Detektor genannt werden. Der 'Verstärker' selbst ist deutlich unwichtiger als die Phasen-Empfindlichkeit. Mit einem Lock-In-Verstärker kann man ein Signal in einem sehr schmalen Frequenzintervall detektieren (ähnlich einem schmalen Bandpass-Filter). Man ist aber zusätzlich auch sensitiv auf die Phasenlage des Signal relativ zu einer Referenzphase.

Motivation

Kontinuierliches Signal

Wir betrachten eine kontinuierlich leuchtende Lichtquelle, deren Leistung wir durch eine Photodiode detektieren und den Photostrom verstärken und messen. Solange das Signal groß ist, geht das problemlos. Wenn das Signal jedoch klein ist, dann spielt das Rauschen des Photodetektors und mögliche Störquellen eine Rolle.

Detektion eines kontinuierlichen Signals

> Skizzieren Sie das zu erwartende Rauschspektrum der Lichtquelle sowie des Photodetektors mit und ohne eingeschaltete Lichtquelle.

Moduliertes Signal

Die Dinge werden etwas besser, wenn wir die Lichtquelle modulieren. Wir können beispielsweise die Stromversorgung periodisch mit der Frequenz f_m ein- und ausschalten, oder einen mechanischen Zerhacker (engl. chopper) verwenden, der periodisch den Strahl unterbricht. Dadurch verschieben wir die interessanten Frequenzen des Signals nach f_m.



Detektion eines modulierten Signals durch Bandpass-Filterung und Gleichrichtung

> Gegeben sei das Frequenz-Spektrum $S(\omega)$ der unmodulierten Lichtquelle (z.B. aus ihrer Skizze oben). Der Chopper multipliziert den zeitlichen Verlauf $s(t)$ mit einer Rechteck-Funktion, die gleich lang den Wert '1' und den Wert '0' zeigt und die Frequenz f_m besitzt. Berechnen Sie das Frequenz-Spektrum $M(\omega)$ nach dem Chopper, was also von der Photodiode detektiert wird.

Die Photodiode detektiert in diesem Fall ein moduliertes Signal. Die interessierende Größe ist die Amplitude des Frequenzkomponente bei f_m. Eine naheliegende Möglichkeit ist, das elektrische Signal der Photodiode durch einen Bandpassfilter bei der Frequenz f_m zu filtern und danach gleichzurichten.

Das ist machbar, hat aber einige Nachteile

- Die Güte $Q = \Delta\omega/\omega$ eines Filters ist limitiert. Bei hohen Frequenzen ω kann also die Bandbreite $\Delta\omega$ nicht beliebig schmal werden. Man würde aber gerne ein sehr kleines $\Delta\omega$ verwenden, um möglichst viel Rauschen zu unterdrücken.
- Falls die Modulationsfrequenz f_m zeitlich driftet, weil beispielsweise der Motor nicht immer exakt gleich schnell läuft, dann liegt das Signal nicht mehr in der Mitte des Filter-Passbandes und es wird abgeschwächt. Frequenz-Drifts des Motors erscheinen also als Amplitudenänderung des Signals
- Ein Gleichrichter liefert nur positive Spannungen. Auch ohne Signal wird daher das immer vorhandene Rauschen um einen positiven Wert zentriert sein, nicht um die Null. Dieser statische Rausch-Beitrag ist von einem kleinen echten Signal nicht zu unterscheiden.

Phasen-empfindliche Detektion

Ein phasen-empfindlicher Detektor löst diese Probleme. Wir betrachten zwei Varianten:

Schalter = Demodulation mit Rechteck

Wir detektieren und verstärken unser Signal weiterhin mit einem Photodetektor. Dann jedoch folgt kein Bandpass-Filter, sondern ein Schalter, mit dem wir periodisch mit der Frequenzen f_r das Vorzeichen des Signal ändern, also mit einer Rechteck-Funktion multiplizieren, die zwischen '1' und '-1' oszilliert. Danach folgt ein Tiefpass-Filter und wir messen die Amplitude des Signals.



Phasenempfindliche Detektion durch Umschalten

Der Trick ist, dass wir die Referenz-Frequenz f_r der Demodulation gleich der der Modulation f_m wählen und zwar phasenstarr. Es geht also ein Kabel vom Chopper und eines von der Diode zum Lock-In-Verstärker.

> Warum ist es nicht möglich, einfach nur beide Frequenz auf den gleichen Wert zu stellen?

Wenn die uns interessierende Frequenzkomponente im Diodensignal in Phase ist mit dem Referenzsignal, dann werden die negativen Halbwellen mit '-1' multipliziert und umgeklappt, so dass nach der Tiefpass-Filterung ein Wert übrig bleibt. Wenn aber beispielsweise kein Signal anliegt, sondern nur Rauschen um die Null herum, dann wird ein Teil des Rauschens invertiert. Dessen Mittelwert ist aber weiterhin zentriert um die Null.

Die Bandbreite dieses 'synchronen Filters' ist durch die Zeitkonstante T des Tiefpass-Filters gegeben. Die kann aber beliebig lang sein, unabhängig von f_m, so dass beliebig schmale Filter erzeugt werden können.

Mixer = Demodulation mit Kosinus

Alternativ kann man an Stelle des Schalters auch einen Mixer verwenden, also ein elektronisches Bauteil, dessen Ausgangssignal dem Produkt seiner beiden Eingangsignale entspricht. Wir erzeugen also aus dem Diodensignal $m(t)$ und einer cosinus-förmigen Referenz-Wellenform $r(t)$ ein Signal

$$u(t) = m(t) \times r(t)$$

> Ausgehend von obigem Diodensignal $M(\omega)$, wie sieht $U(\omega)$ aus?

Analog kann man auch obige Schalter-Variante in Frequenzraum beschreiben.



Phasenempfindliche Detektion durch Mischen

Typischerweise demoduliert man sowohl mit einem Kosinus als auch einen Sinus (das nennt man manchmal 'dual phase'), und erzeugt nach Tiefpass-Filterung zwei Signale $x(t)$ und $y(t)$. Aus diesen kann man dann die Amplitude $R = \sqrt{x^2 + y^2}$ und den Phasenwinkel $\phi = \mathbf{atan}(y/x)$ berechnen.

> Sei
>
> $$m(t) = a\cos(f_m t + \phi_m)$$
>
> und
>
> $$r(t) = \cos(f_r t + \phi_r) \quad \mathbf{bzw.} \quad \sin(\dots)$$
>
> Geben Sie eine Gleichung für das detektierte Signal $x(t)$ und $y(t)$ an.

Frage

> Wenn man mit einem Chopper den Lichtstrahl moduliert, ist es dann geschickter mit einem Rechteck oder einem Cosinus zu demodulieren? In welchen Fällen ist dies andersrum?

Simulator

Einen Lock-In-Verstärker kann man mit reiner analoger Schaltungstechnik aufbauen (und das wird auch noch so verkauft). Oder man digitalisiert das Signal mit einem Analog-zu-Digital-Wandler und verarbeitet alles digital. In beiden Fällen arbeitet der LIA kontinuierlich. Zu jedem Zeitschritt wird ein Wert eingelesen und (ggf. mit etwas Verzögerung) der zugehörige demodulierte Wert ausgegeben.

Im Folgenden bespreche ich einen Simulator, der der Einfachheit halber auf ganzen Datensätzen arbeitet, dieses Wissen um die Zukunft des Signals aber nicht benutzt.

Wir definieren den zeitlichen Abstand dt der Messwerte und eine Zeitachse.

```
begin
    dt = 0.001
    t = (0:dt:10)
end;
```

Wir erzeugen ein Kosinus-förmiges Signal mit der Frequenz f_0, der Phase $d\phi$ der Amplitude A, das wir aber erst bei $t = 1$s einschalten. Zusätzliche gibt es weißes (= spektral flaches) Rauschen mit der Standard-Abweichung 1. Die Photodiode detektiert die Summe von beiden.

```
begin
    f0 = 10
    dϕ = 0.5
    A = zeros(length(t))
    A[t .> 1] .= 2    # switch on at t=1 sec

    signal = A .* (1 .+ cos.(2π * t * f0 .+ dϕ))
    rauschen = randn(length(signal))
    dunkelstrom = 1

    diode = signal .+ rauschen .+ dunkelstrom
end;
```




```
plot(t[t .< 2], diode[t .< 2],
    xlabel="Zeit (s)", ylabel="Diodenstrom (arb.u.)",
    legend= false, title="Die ersten 2 Sekunden der Messung")
```

Jetzt bauen wir einen Lock-In-Verstärker.

Zunächst erzeugen wir zwei um 90 Grade phasenverschobene Referenz-Wellenformen bei der Frequenz f_0 von oben.




```
begin
    refx = cos.(2π * t * f0)
    refy = sin.(2π * t * f0)

    plot(t[t .< 2], refx[t .< 2])
    aside(embed_display(plot!(t[t .< 2], refy[t .< 2],
        leg=false, xlabel="Zeit", ylabel="ref")))
end
```

Dann mischen wir das Diodensignal mit den Referenz-Wellenformen




```
begin
    xm = diode .* refx
    ym = diode .* refy

    plot(t[t .< 2], xm[t .< 2])
    aside(embed_display(plot!(t[t .< 2], ym[t .< 2],
        leg=false, xlabel="Zeit", ylabel="misch")))
end
```

Schließlich nutzen wir einen Tiefpass-Filter aus 'DSP'. Der Faktor 2 korrigiert den Leistungsverlust durch das Abschneiden der Summenfrequenzen.

```
begin
    using DSP
    f_LP = 2
    myfilter = digitalfilter( Lowpass(f_LP; fs= 1/dt), Butterworth(4))
    x = 2 .* filt(myfilter, xm)
    y = 2 .* filt(myfilter, ym)
end;
```

Insgesamt sieht das gemessene Signal dann so aus




```
begin
    plot(t,x, label="x")
    plot!(t,y, label="y", xlabel="Zeit (s)", ylabel="Ausgang des LIA")
    plot!(t, A, label="A orig.")
end
```

Die Werte von x und y erreichen A nicht, weil wir eine Phasenverschiebung $d\phi$ vorgegeben hatten. Wir berechnen jetzt also R und ϕ




```julia
begin
    R = sqrt.(x.^2 + y.^2)
    phi = -1 .* atan.(y,x)

     plot(t,R, label="R", layout=(2,1))
    plot!(t,A, label="A orig.", xlabel="Zeit (s)", ylabel="Amplitude")
    plot!(t, phi, label="phi", subplot=2)
    plot!(t, dϕ .+ t .* 0, label="phi orig.", subplot=2, ylabel="Phase (rad)")
end
```

Rauschen

In dem hier simulierten Fall haben wir das Rauschen als konstant in Frequenzraum angenommen, mit $\sigma = 1$

```
1.0076123047678502
```

```julia
std(rauschen)
```

Die Sample-Rate dt ist 1ms, die Nyquist-Frequenz also 500 Hz. Weil gelten muss

$$\int_{f>0} |\mathrm{Rauschen}(f)|^2 \, df = \sigma^2$$

ist also die spektrale Rauschdichte

$$\mathrm{Rauschen}(f) = \frac{\sigma}{\sqrt{500 \, Hz}} = \mathrm{const.}$$

Unser Filter hat eine NEBW von 2.1 Hz

```
2.0602906824956917
```

```
begin
    H = freqresp(myfilter, (0:0.0001:pi))
    df_filter_resp = 1 ./ (2 * length(H) * dt)
    nebw = sum(abs.(H).^2) * df_filter_resp
end
```

Durch das Tiefpass-Filtern sollte das Rauschen also um den Faktor $\sqrt{NEBW/f_{\mathrm{Nyq.}}}$ zurück gehen, was auch so geschieht:

```
fnyq= 1/ (2 * dt);
```

```
(0.062895, 0.0646804)
```

```
(std(filt(myfilter, rauschen)), std(rauschen) * sqrt(nebw / fnyq) )
```

Das Rauschen des LIA-Ausgangs ist durch die Rauschdichte im Filter-Passband gegeben, also auch um den Faktor $\sqrt{NEBW/f_{\mathrm{Nyq.}}}$ reduziert. Wenn man R statt x oder y betrachtet, kommt durch Fehlerfortpflanzung noch ein Faktor $\sqrt{2}$ dazu. (Wir schneiden hier die ersten 5 Sekunden ab.)

```
1.522287119857451
```

```
std(R[t .> 5]) / sqrt(nebw / fnyq)
```

Boxcar averaging

Alternativ zu einem Lock-In-Verstärker kann man auch boxcar averaging machen. Ein gepulster Laser produziert kurze Lichtpulse (ca. ps bis ns) in periodischen Abständen (ca. kHz). Dann kann man den Photostrom der Photodiode über kurze Intervalle entsprechend der Pulslänge integrieren, und in den langen Intervallen zwischen den Laserpulsen ignorieren. Dies tut ein boxcar averager.

Man kann dies als Variante des Lock-in-Verstärker sehen, auch wenn es technologisch anders gelöst ist. Im Prinzip aber demoduliert man das Signal mit einer Rechteck-Referenzwelle, nur dass diese jetzt nicht wie oben gleich lang 1 und -1 zeigt, sondern die '1'-Phasen viel kürzer sind.

Technisch integriert man das Signal während der '1'-Phasen und mittelt dann über viele dieser Phasen. Die '-1'-Phasen ignoriert man oder beschafft sich eine andere Hintergrund-Messung.

Beispiel SR 830

Lock-In Verstärker SR830 von Stanford Research

- Manual
- website

```
using PlutoUI
```

```
using Plots, StatsBase, FFTW
```

Inhalt


```
TableOfContents(title="Inhalt")
```

```
aside(x) = PlutoUI.ExperimentalLayout.aside(x);
```